

AVISOS - Avaliação em Época Normal

[Expand]

Material de Uso Obrigatório

[Expand]

Contents

- 1 Tipos de Dados
- 2 Manipulação de Nomes
 - 2.1 Espaço de nomes e visibilidade dos identificadores
 - 2.2 Validade das variáveis
- 3 Convenções Lexicais
 - 3.1 Estrutura de linha (linhas lógicas)
 - 3.1.1 Linhas físicas
 - 3.1.2 Marcador de continuação: junção explícita de linhas físicas
 - 3.2 Caracteres brancos
 - 3.2.1 Linhas brancas
 - 3.2.2 Indentação
 - 3.3 Comentários
 - 3.4 Palavras-chave
 - 3.5 Tipos
 - 3.6 Operadores de expressões
 - 3.7 Delimitadores e terminadores
 - 3.8 Identificadores (nomes)
 - 3.9 Literais
 - 3.9.1 Inteiros
 - 3.9.2 Reais em vírgula flutuante
 - 3.9.3 Cadeias de caracteres
 - 3.9.4 Ponteiros
- 4 Gramática
 - 4.1 Tipos, identificadores, literais e definição de expressões
 - 4.2 Left-values
 - 4.3 Ficheiros
 - 4.4 Declaração de variáveis
 - 4.5 Símbolos globais
 - 4.6 Inicialização
- 5 Funções
 - 5.1 Declaração
 - 5.2 Invocação
 - 5.3 Corpo
 - 5.4 Função principal e execução de programas
- 6 Instruções
 - 6.1 Blocos
 - 6.2 Instrução condicional
 - 6.3 Instrução de iteração
 - 6.4 Instrução de terminação
 - 6.5 Instrução de continuação
 - 6.6 Instrução de retorno
 - 6.7 Expressões como instruções

- 6.8 Instruções de impressão
- 7 Expressões
 - 7.1 Expressões primitivas
 - 7.1.1 Identificadores
 - 7.1.2 Leitura
 - 7.1.3 Parênteses curvos
 - 7.1.4 Funções
 - 7.2 Expressões resultantes de avaliação de operadores
 - 7.2.1 Indexação de ponteiros
 - 7.2.2 Identidade e simétrico
 - 7.2.3 Reserva de memória
 - 7.2.4 Expressão de indicação de posição
 - 7.2.5 Expressão de dimensão
- 8 Exemplos e Testes
- 9 Omissões e Erros

L22 é uma linguagem imperativa não estruturada. Um dos objectivos da linguagem é permitir a fácil leitura, pelo que é fortemente restritiva relativamente ao aspecto e à posição do código nas linhas (indentação).

Este manual apresenta características básicas da linguagem (tipos de dados, manipulação de nomes); convenções lexicais; estrutura/sintaxe; especificação das funções; semântica das instruções; semântica das expressões; e, finalmente, alguns exemplos.

Tipos de Dados

A linguagem é fracamente tipificada (são efectuadas algumas conversões implícitas). Existem 4 tipos de dados básicos, todos compatíveis com a linguagem C, e com alinhamento em memória sempre a 32 bits:

- Tipos numéricos: os inteiros, em complemento para 2, ocupam 4 bytes; os reais, em vírgula flutuante, ocupam 8 bytes (IEEE 754).
- As cadeias de caracteres são vectores de caracteres terminados por ASCII NUL (carácter com o valor zero). Variáveis e literais deste tipo só podem ser utilizados em atribuições, impressões, ou como argumentos/retornos de funções.
- Os ponteiros representam endereços de objectos e ocupam 4 bytes. Podem ser objecto de operações aritméticas (deslocamentos) e permitem aceder ao valor apontado.

Os tipos suportados por cada operador e a operação a realizar são indicados na definição das expressões.

Existem ainda tipos associado a valores funcionais, i.e., tipos que descrevem a interface de funções (ver abaixo). Os valores em memória associados a estes tipos são efectivamente ponteiros, mas para funções e não para dados, podendo ser usados para invocar as funções correspondentes. Estes ponteiros não aceitam operações de aritmética de ponteiros ou de indexação (embora ponteiros para estes ponteiros as aceitem).

Manipulação de Nomes

Os nomes (identificadores) correspondem exclusivamente a variáveis. As funções são referenciadas através de ponteiros nomeados. Nos pontos que se seguem, usa-se o termo entidade para as designar indiscriminadamente, explicitando-se quando a descrição for válida apenas para um dos casos. Existem ainda nomes para funções externas: neste caso, os nomes referem directamente o código dessas funções (à la C; ver **foreign** abaixo).

Espaço de nomes e visibilidade dos identificadores

O espaço de nomes global é único, pelo que um nome utilizado para designar uma entidade num dado contexto não pode ser utilizado para designar outras (ainda que de natureza diferente).

Os identificadores são visíveis desde a declaração até ao fim do alcance: ficheiro (globais) ou bloco (locais). A reutilização de identificadores em contextos inferiores encobre declarações em contextos superiores: redeclarações locais podem encobrir as globais até ao fim de um bloco. É possível utilizar símbolos globais nos contextos dos blocos das funções, mas não é possível declará-los (ver símbolos globais).

Validade das variáveis

As entidades globais (declaradas fora de qualquer função), existem durante toda a execução do programa. As variáveis locais a uma função existem apenas durante a sua execução. Os argumentos formais são válidos enquanto a função está activa.

Convenções Lexicais

Para cada grupo de elementos lexicais (tokens), considera-se a maior sequência de caracteres constituindo um elemento válido. Assim, por exemplo, a designação `>=` é sempre um único elemento lexical (por oposição à situação ilegal de se terem dois símbolos: `>` seguido de `=`).

Estrutura de linha (linhas lógicas)

Um programa está dividido em linhas lógicas e uma instrução não pode ocupar mais de uma linha lógica, excepto se a instrução permitir explicitamente mudanças de linha através, por exemplo, da utilização de blocos. Uma linha lógica pode ser constituída por uma ou mais linhas físicas separadas pelo marcador de continuação: `...` (três pontos seguidos).

Linhas físicas

Uma linha física pode ter qualquer comprimento. A terminação de uma linha física depende da convenção utilizada pelo sistema operativo de suporte utilizado. Assim, em UNIX (o ambiente considerado neste projecto), o carácter utilizado é o ASCII 0x0A (linefeed). Embora sistemas diferentes possam considerar outros caracteres para marcar o final de linha, neste caso, apenas se considera o mencionado acima.

Marcador de continuação: junção explícita de linhas físicas

Duas ou mais linhas físicas podem ser juntas numa única linha lógica através da utilização do marcador de continuação: `...` (três pontos seguidos), quando ocorrem imediatamente antes da terminação da linha física (ver exemplo -- neste exemplo, a condição está dividida em várias linhas físicas).

```
if (year > 2006 and not year > 2010 and not day < 1 ...
    and ((month == 1 or month == 3 or month == 5 or month == 7 ...
        or month == 8 or month == 10 or month == 12) and not day > 31 ...
        or (month == 4 or month == 6 or month == 9 or month == 11) and not day > 36 ...
        or (month == 8 or month == 10 or month == 12) and not day > 31 ...
        or month == 2 and (year == 2008 and not day > 29 or not year == 2008 ...
            and not day > 28))) then:
    writeln "great day!"
else:
    writeln "not a great day..."
```

O marcador de continuação pode também ser utilizado em cadeias de caracteres literais, sendo ilegal em qualquer outro contexto. Em particular, não pode ser utilizado para permitir a continuação de tokens (identificadores, palavras-chave, delimitadores, operadores, literais inteiros ou reais), nem comentários explicativos.

Caracteres brancos

São considerados caracteres brancos aqueles que, embora servindo para separar os elementos lexicais, não representam nenhum elemento lexical. São considerados caracteres brancos os designados pelo código ASCII 0x20 (espaço). Note-se que os caracteres de tabulação horizontal (ASCII 0x09) não são considerados brancos e não devem ocorrer fora das cadeias de caracteres.

Note-se que os caracteres mencionados acima, apesar de brancos, têm significado especial se aparecerem no início de uma linha lógica. Analogamente, os caracteres de mudança de linha também têm significado especial (ver linhas brancas e indentação),

Linhas brancas

Uma linha lógica que contenha apenas caracteres brancos e comentários é ignorada, não sendo gerado nenhum elemento lexical nem alterada a indentação.

Indentação

Os caracteres espaço não são considerados brancos no início de uma linha lógica. Neste caso, as suas cadeias determinam o nível de indentação do código.

A indentação não pode continuar na linha física seguinte mediante a utilização de uma junção de linhas. Assim, no máximo, o número de espaços brancos até ao primeiro carácter de continuação determina o nível de indentação.

Embora os diversos níveis de indentação possam ser diferentes, o significado pode ser o mesmo (ver exemplo seguinte).

(a) OK: bom estilo

```
var max = (int a, int b) -> int:
  if (a > b) then:
    return a
  else:
    return b
```

(b) OK: mau estilo

```
var max = (int a, int b) -> int:
  if (a > b) then:
    return a
  else:
    return b
```

Errado!

```
var max = (int a, int b) -> int:
  if (a > b) then:
    return a
  else:
    return b
```

Comentários

Existem dois tipos de comentários, que também funcionam como elementos separadores:

- **explicativos** -- começam com ; e acabam no fim da linha física; e
- **operacionais** -- começam com (* e terminam com *), podendo estar aninhados.

Se as sequências de início fizerem parte de uma cadeia de caracteres, não iniciam um comentário (ver definição das cadeias de caracteres).

Palavras-chave

As seguintes palavras-chave são reservadas, não constituindo identificadores (devem ser escritas exactamente como indicado):

- tipos: **int double text void**
- declarações: **foreign use public var**
- operadores: **and or not**
- instruções: **if then: elif else: while do: stop again write writeln return**
- expressões: **input null**
- outras: **begin end**

Note-se que **do**, **then**, e **else** não são palavras-chave.

Tipos

Os seguintes elementos lexicais designam tipos em declarações (ver gramática): **int** (inteiro), **double** (real), **text** (cadeia de caracteres).

Os tipos correspondentes a ponteiros são outros tipos delimitados por [e], designando uma indirecção e não o objecto directo (ver gramática).

Operadores de expressões

São considerados operadores os elementos lexicais apresentados na definição das expressões.

Delimitadores e terminadores

Os seguintes elementos lexicais são delimitadores/terminadores: , (vírgula) e (e) (delimitadores de expressões).

Identificadores (nomes)

São iniciados por uma letra, seguindo-se 0 (zero) ou mais letras, dígitos, ou _ (sublinhado). O comprimento do nome é ilimitado e dois nomes são distintos se houver alteração de maiúscula para minúscula, ou vice-versa, de pelo menos um carácter.

Literais

São notações para valores constantes de alguns tipos da linguagem (não confundir com constantes, i.e., identificadores que, em algumas linguagens, designam elementos cujo valor não pode ser alterado durante a execução do programa).

Inteiros

Um literal inteiro é um número não negativo. Números negativos podem ser construídos pela aplicação do operador de negação unária (-) a um literal (sempre positivo).

Literais inteiros decimais são constituídos por sequências de 1 (um) ou mais dígitos de **0** a **9**.

Literais inteiros em base 7 começam sempre pelo dígito 0 (zero), sendo seguidos de um ou mais dígitos de **0** a **6** (note-se que **09** é um literal inválido em base 7). Exemplo: **006**.

Se não for possível representar um literal Inteiro na máquina, devido a um overflow, deverá ser gerado um erro lexical.

Reais em vírgula flutuante

Os literais reais (sempre positivos) são expressos tal como em C (exclusivamente em base 10).

Não existem literais negativos (números negativos resultam da operação unária -).

Um literal sem . (ponto decimal) nem parte exponencial é do tipo inteiro.

Exemplos: **3.14**, **1E3** = 1000 (número inteiro representado em virgula flutuante). **12.34e-24** = 12.34×10^{-24} (notação científica).

Cadeias de caracteres

As cadeias de caracteres são delimitadas por aspas (") e podem conter quaisquer caracteres, excepto ASCII NUL (0x00) e ASCII LF (0x0A). Nas cadeias, os delimitadores de comentários não têm significado especial. Se for escrito um literal que contenha **\0**, então a cadeia termina nessa posição. Exemplo: **ab\0xy** tem o mesmo significado que **ab**.

É possível designar caracteres por sequências especiais (iniciadas por \), especialmente úteis quando não existe representação gráfica directa. As sequências especiais correspondem aos caracteres ASCII HT, LF e CR (**\t**, **\n** e **\r**, respectivamente), aspa (**\"**). *backslash* (****), ou a quaisquer outros especificados através de 1 a 3 dígitos em base 7, designando valores de 8 bits (e.g., **\013** ou apenas **\13** se o carácter seguinte não representar um dígito em base 7). Exemplo: **xy\013z** tem o mesmo significado que **xy\13z** e que **xy\1nz**.

Elementos lexicais distintos que representem duas ou mais cadeias consecutivas são representadas na linguagem como uma única cadeia que resulta da concatenação. Exemplo: **"ab" "cd"** é o mesmo que **"abcd"**.

Ponteiros

O único literal admissível para ponteiros corresponde ao ponteiro nulo e é indicado pela palavra-chave **null**. Este literal é compatível com todos os tipos de ponteiro. Os inteiros não são convertíveis em ponteiros, pelo que o valor **0** (zero) não é um valor inicial admissível para ponteiros.

Gramática

A gramática da linguagem está resumida abaixo. Considerou-se que os elementos em tipo fixo são literais, que os parênteses curvos agrupam elementos, que elementos alternativos são separados por uma barra vertical, que elementos opcionais estão entre parênteses rectos, que os elementos que se repetem zero ou mais vezes estão entre $\langle \text{ e } \rangle$. Alguns elementos usados na gramática também são elementos da linguagem descrita se representados em tipo fixo (e.g., parênteses).

<i>ficheiro</i>	→	$\langle \text{ declaração } \rangle$ [<i>programa-principal</i>]
<i>declaração</i>	→	[<i>qualificador</i>] <i>tipo identificador</i> [= <i>expressão</i>]
	→	[<i>qualificador</i>] [var] <i>identificador</i> = <i>expressão</i>
<i>programa-principal</i>	→	begin <i>bloco</i> end
<i>função</i>	→	() -> <i>tipo</i> : <i>bloco</i>
	→	(<i>variáveis</i>) -> <i>tipo</i> : <i>bloco</i>
<i>variáveis</i>	→	<i>variável</i> $\langle \text{ , } \text{ variável } \rangle$
<i>tipo</i>	→	int double text [<i>tipo</i>] <i>tipo-de-função</i>
<i>tipo-de-função</i>	→	<i>tipo</i> < $\langle \text{ tipo } \rangle$ >
<i>bloco</i>	→	$\langle \text{ declaração } \rangle$ $\langle \text{ instrução } \rangle$
<i>instrução</i>	→	<i>expressão</i> write <i>expressões</i> writeln <i>expressões</i>
	→	again stop return [<i>expressão</i>]
	→	<i>instrução-condicional</i> <i>instrução-de-iteração</i> <i>bloco</i>
<i>instrução-condicional</i>	→	if (<i>expressão</i>) then : <i>bloco</i> $\langle \text{ elif (expressão) then: bloco } \rangle$ [else : <i>bloco</i>]
<i>instrução-de-iteração</i>	→	while (<i>expressão</i>) do : <i>bloco</i>
<i>expressões</i>	→	<i>expressão</i> $\langle \text{ , } \text{ expressão } \rangle$

Tipos, identificadores, literais e definição de expressões

Algumas definições foram omitidas da gramática: tipos de dados, qualificadores (ver declarações de variáveis), identificador (ver identificadores), literal (ver literais); expressão (ver expressões). Note-se que *função* é qualquer especificação de função (corpo ou ponteiro com o tipo apropriado). Neste sentido, as funções contam como expressões primitivas.

Quanto a tipos de dados, **int** designa valores inteiros, **double** designa valores reais, **text** designa cadeias de caracteres. Os ponteiros são tipos compostos por um dos tipos básicos entre parênteses rectos, e.g., **[int]**.

Ponteiros para funções são definidos a partir dos tipos de dados anteriormente descritos e do tipo especial **void** (ver a seguir). O tipo é indicado com o formato indicado na gramática, i.e., tipo de retorno seguido dos tipos dos argumentos (zero ou mais separados por vírgulas).

Um ponteiro declarado com um tipo de função indica o endereço da função correspondente. Estes ponteiros não suportam aritmética de ponteiros.

O tipo **void** apenas pode ser usado para indicar a ausência de retorno ou para declarar um ponteiro genérico. Neste caso, o aninhamento é irrelevante, i.e., **[void]** e **[[[void]]]** são equivalentes. Um ponteiro deste tipo é compatível com todos os outros tipos de ponteiros. A aritmética de ponteiros decrementa/incrementa em uma unidade o valor de um ponteiro do tipo **[void]**.

Left-values

Os *left-values* são posições de memória que podem ser modificadas (excepto onde proibido pelo tipo de dados). Os elementos de uma expressão que podem ser utilizados como left-values encontram-se individualmente identificados na semântica das expressões.

Ficheiros

Um ficheiro é designado por principal se contiver a função principal (onde se inicia o programa).

Declaração de variáveis

Uma declaração de variável indica sempre um tipo de dados (implícito ou explícito) e um identificador.

Exemplos:

- Inteiro: **int i**
- Real: **double r**
- Cadeia de caracteres: **text s**
- Ponteiro para inteiro: **[int] p1** (equivalente a **int*** em C)
- Ponteiro para real: **[double] p2** (equivalente a **double*** em C)
- Ponteiro para cadeia de caracteres: **[text] p3** (equivalente a **char**** em C)
- Ponteiro para ponteiro para inteiro: **[[int]] p4** (equivalente a **int**** em C)

É possível usar o pseudo-tipo **var** se a declaração tiver um valor inicial: neste caso, o tipo é o do valor inicial.

Exemplo:

- Variável inteira: **var i = 1**
- Variável real: **var f = 2.0**
- etc.

Símbolos globais

Por omissão, os símbolos são privados a um módulo, não podendo ser importados por outros módulos.

A palavra-chave **public** permite declarar um identificador como público, tornando-o acessível a partir de outros módulos.

Quando usado com a palavra-chave **var**, esta é opcional. Note-se que a declaração de uma variável tem de ter sempre, ou o qualificador, ou o tipo, podendo estar ambos presentes.

A palavra-chave **use** permite declarar num módulo variáveis definidas noutros módulos. Neste caso, não pode ser especificado o valor inicial dessas variáveis, pelo que não é possível o uso de **var** em conjunto com **use**.

A palavra-chave **foreign** deve ser usada para declarar símbolos de função com uma convenção de chamada diferente da da linguagem L22, e.g. para importar funções definidas em C. Além de poderem ser usados para chamar as funções que designam, os símbolos assim declarados podem ser atribuídos a variáveis com o tipo apropriado.

Exemplos:

- Declarar variável privada ao módulo: **double d1 = 22.0**
- Declarar variável pública: **public double d2 = 7.0**
- Declarar variável pública: **public var d2 = 7.0** (igual à anterior)
- Declarar variável pública: **public d2 = 7.0** (igual à anterior)
- Usar definição externa de variável pública: **use double d2**
- Usar ponteiro para função (nativa L22) definida externamente: **use int<int> factorial**
- Usar função definida externamente (e.g. em C): **foreign [void]<int> malloc**

Inicialização

Quando existe, é a designação do objecto que segue o símbolo **=**: inteiro, real, cadeia de caracteres, ponteiro ou função.

Entidades reais podem ser inicializadas por expressões inteiras (conversão implícita). A expressão de inicialização deve ser um literal se a variável for global. A associação de valores funcionais a variáveis pode ser realizada quando os tipos forem

covariantes.

A palavra **var** podem ser usadas em lugar do tipo para declarar uma variável. Quando usada, o tipo da variável é inferido a partir do valor inicial (nestes casos, o valor inicial é obrigatório).

Exemplos:

- Inteiro (literal): **int i = 3**
- Inteiro (expressão): **int i = j + 1**
- Real (literal): **double r = 3.2**
- Real (expressão): **double r = i - 2.5**
- Cadeia de caracteres (literal): **text s = "olá"**
- Cadeia de caracteres (literais): **text s = "olá" "mãe"**
- Ponteiro (literal): **[[[double]]] p = null**
- Ponteiro (expressão): **[int] p = q + 1**
- Função:

```
int<int> f1
int<double> g1
double<int> g2
int<int> f2 = f1 ;; mesmo tipo
f2 = g1 ;; ok: tipos covariantes
f2 = g2 ;; ERRADO
```

Funções

Uma função permite agrupar um conjunto de instruções num corpo, executado com base num conjunto de parâmetros (os argumentos formais), quando é invocada a partir de uma expressão.

Declaração

As funções são anónimas, mas sempre referenciadas por identificadores ou ponteiros, tal como outros tipos de dados. O tipo de retorno de uma função que não produz valores de retorno é **void**.

As funções que recebam argumentos devem indicá-los no cabeçalho. Funções sem argumentos definem um cabeçalho vazio. Os qualificadores de exportação/importação **public** ou **use** (ver símbolos globais) são aplicados às variáveis usadas para referir as funções. Não é possível aplicá-los às declarações dos argumentos de uma função. Não é possível especificar valores por omissão para os argumentos de uma função, nem para o valor de retorno.

A declaração de uma variável do tipo função sem ser iniciada é utilizada para caracterizar um identificador exterior ou para efectuar declarações antecipadas (utilizadas para pré-declarar funções que sejam usadas antes de ser definidas, por exemplo, entre duas funções mutuamente recursivas).

Invocação

Uma função pode ser invocada através de qualquer expressão (ponteiro) do tipo apropriado que refira essa função (ponteiro não nulo). O símbolo **@** pode ser usado dentro da própria função para efectuar uma invocação recursiva. Não é possível o uso de **@** no programa principal.

Se existirem argumentos, na invocação da função, o ponteiro é seguido de uma lista de expressões delimitadas por parênteses curvos. Esta lista é uma sequência, possivelmente vazia, de expressões separadas por vírgulas. O número e tipo de parâmetros actuais deve ser igual ao número e tipo dos parâmetros formais da função invocada (a menos de conversões implícitas). A ordem dos parâmetros actuais deverá ser a mesma dos argumentos formais da função a ser invocada.

De acordo com a convenção Cdecl, a função chamadora coloca os argumentos na pilha e é responsável pela sua remoção, após o retorno da chamada. Assim, os parâmetros actuais devem ser colocados na pilha pela ordem inversa da sua declaração (i.e., são avaliados da direita para a esquerda antes da invocação da função e o resultado passado por cópia/valor). O endereço de retorno é colocado no topo da pilha pela chamada à função.

Corpo

O corpo de uma função consiste num bloco que contém declarações (opcionais) seguidas de instruções (opcionais). Não é possível aplicar os qualificadores de exportação (**public**) ou de importação (**use** ou **foreign**) (ver símbolos globais) dentro do corpo de uma função.

Uma instrução **return** causa a interrupção da função. O valor devolvido por uma função, através de expressão usada como argumento da instrução **return**, deve ser do tipo declarado no cabeçalho da função.

É um erro especificar um valor de retorno se a função for declarada como não retornando um valor (indicada como **void**).

Qualquer bloco (usado, por exemplo, numa instrução condicional ou de iteração) pode definir variáveis, incluindo outras funções. Funções definidas dentro de um bloco não têm acesso às variáveis em contexto na função actual onde ocorrem.

Função principal e execução de programas

Um programa é a sequência de instruções que seguem as declarações globais num ficheiro, delimitadas pelas palavras-chave **begin** e **end**. Esta sequência forma o que em algumas linguagens se chama a função principal. Os argumentos com que o programa foi chamado podem ser obtidos através de funções **argc** (devolve o número de argumentos); **argv** (devolve o n-ésimo argumento como uma cadeia de caracteres, com $n > 0$); e **envp** (devolve a n-ésima variável de ambiente como uma cadeia de caracteres, com $n > 0$). Apenas um dos módulos do programa pode definir a função principal, i.e., se existirem vários módulos, apenas um deles pode conter mais do que declarações de variáveis globais.

```
foreign int<> argc
foreign text<int> argv
foreign text<int> envp
```

O valor de retorno da função principal é devolvido ao ambiente que invocou o programa. Este valor de retorno segue as seguintes regras (sistema operativo): 0 (zero), execução sem erros; 1 (um), argumentos inválidos (em número ou valor); 2 (dois), erro de execução. Os valores superiores a 128 indicam que o programa terminou com um sinal. Em geral, para correcto funcionamento, os programas devem devolver 0 (zero) se a execução foi bem-sucedida e um valor diferente de 0 (zero) em caso de erro.

A biblioteca de run-time (RTS) contém informação sobre outras funções de suporte disponíveis, incluindo chamadas ao sistema (ver também o manual da RTS).

Instruções

Excepto quando indicado, as instruções são executadas em sequência.

Blocos

Cada bloco tem uma zona de declarações de variáveis locais (facultativa), seguida por uma zona com instruções (possivelmente vazia). É possível declarar e definir funções dentro de blocos.

A visibilidade das variáveis é limitada ao bloco em que foram declaradas. As entidades declaradas podem ser directamente utilizadas em sub-blocos ou passadas como argumentos para funções chamadas dentro do bloco. Caso os identificadores usados para definir as variáveis locais já estejam a ser utilizados para definir outras entidades ao alcance do bloco, o novo identificador passa a referir uma nova entidade definida no bloco até que ele termine (a entidade previamente definida continua a existir, mas não pode ser directamente referida pelo seu nome). Esta regra é também válida relativamente a argumentos de funções (ver corpo das funções).

Instrução condicional

Esta instrução tem comportamento semelhante ao da instrução **if-else** em C. As partes correspondentes a **elif** comportam-se como encadeamentos de instruções condicionais na parte **else** de um **if-else** à la C.

Instrução de iteração

Esta instrução tem comportamento idêntico ao da instrução **while** em C.

Instrução de terminação

Indicada pela palavra-chave **stop** (quando existe, é a última instrução do seu bloco): termina o ciclo mais interior em que a instrução se encontrar, tal como a instrução **break** em C. Esta instrução só pode existir dentro de um ciclo.

Instrução de continuação

Indicada pela palavra-chave **again** (quando existe, é a última instrução do seu bloco): reinicia o ciclo mais interior em que a instrução se encontrar, tal como a instrução **continue** em C. Esta instrução só pode existir dentro de um ciclo.

Instrução de retorno

A instrução **return**, se existir, é a última instrução do seu bloco. Ver comportamento na descrição do corpo de uma função.

Expressões como instruções

As expressões utilizadas como instruções são avaliadas, mesmo que não produzam efeitos secundários.

Instruções de impressão

As palavras-chave **write** e **writeln** podem ser utilizadas para apresentar valores na saída do programa. A primeira apresenta o argumento sem mudar de linha; a segunda apresenta o argumento mudando de linha. Quando existe mais de uma expressão, as várias expressões são apresentadas sem separação. Valores numéricos (inteiros ou reais) são impressos em decimal. As cadeias de caracteres são impressas na codificação nativa. Ponteiros não podem ser impressos.

Expressões

Uma expressão é uma representação algébrica de uma quantidade: todas as expressões têm um tipo e devolvem um valor.

Existem expressões primitivas e expressões que resultam da avaliação de operadores.

A tabela seguinte apresenta as precedências relativas dos operadores: é a mesma para operadores na mesma linha, sendo as linhas seguintes de menor prioridade que as anteriores. A maioria dos operadores segue a semântica da linguagem C (excepto onde explicitamente indicado). Tal como em C, os valores lógicos são 0 (zero) (valor falso), e diferente de zero (valor verdadeiro).

Tipo de Expressão	Operadores	Associatividade	Operandos	Semântica
primária	() []	não associativos	-	parênteses curvos, indexação, reserva de memória
unária	+ - ?	não associativos	-	identidade e simétrico, indicação de posição
multiplicativa	* / %	da esquerda para a direita	inteiros, reais	C (% é apenas para inteiros)
aditiva	+ -	da esquerda para a direita	inteiros, reais, ponteiros	C: se envolverem ponteiros, calculam: (i) deslocamentos, i.e., um dos operandos deve ser do tipo ponteiro e o outro do tipo inteiro; (ii) diferenças de ponteiros, i.e., apenas quando se aplica o operador - a dois ponteiros do mesmo tipo (o resultado é o número de objectos do tipo apontado entre eles). Se a memória não for contígua, o resultado é indefinido.
comparativa	< > <= >=	da esquerda para a direita	inteiros, reais	C
igualdade	== !=	da esquerda para a direita	inteiros, reais, ponteiros	C
"não" lógico	not	não associativo	inteiros	C

"e" lógico	and	da esquerda para a direita	inteiros	C: o 2º argumento só é avaliado se o 1º não for falso.
"ou" lógico	or	da esquerda para a direita	inteiros	C: o 2º argumento só é avaliado se o 1º não for verdadeiro.
atribuição	=	da direita para a esquerda	todos os tipos	O valor da expressão do lado direito do operador é guardado na posição indicada pelo <i>left-value</i> (operando esquerdo do operador). Podem ser atribuídos valores inteiros a <i>left-values</i> reais (conversão automática). Nos outros casos, ambos os tipos têm de concordar.

Expressões primitivas

As expressões literais e a invocação de funções foram definidas acima.

Identificadores

Um identificador é uma expressão se tiver sido declarado. Um identificador pode denotar uma variável.

Um identificador é o caso mais simples de um *left-value*, ou seja, uma entidade que pode ser utilizada no lado esquerdo (*left*) de uma atribuição.

Leitura

A operação de leitura de um valor inteiro ou real pode ser efectuada pela expressão indicada pela palavra-chave **input**, que devolve o valor lido, de acordo com o tipo esperado (inteiro ou real). Caso se use como argumento dos operadores de impressão ou noutras situações que permitam vários tipos (**write** ou **writeln**), deve ser lido um inteiro.

Exemplos: **a = input** (leitura para **a**), **f(input)** (leitura para argumento de função), **write input** (leitura e impressão), **@(input)** (chamada recursiva a função com argumento que é lido da entrada).

Parênteses curvos

Uma expressão entre parênteses curvos tem o valor da expressão sem os parênteses e permite alterar a prioridade dos operadores. Uma expressão entre parênteses não pode ser utilizada como *left-value* (ver também a expressão de indexação).

Funções

As funções (ponteiros ou o seu código: não confundir com chamadas a funções) podem ser usadas como expressões tipificadas como funções, i.e., ponteiros para funções (mesmo quando não se usa explicitamente um ponteiro).

Exemplo:

```
var f = (int i) -> int:
    return i + 1

var g = (int n, int<int> fun) -> int:
    return fun(n)

begin
    g(3, f) ;; retorna 4
    g(3, (int i) -> int:
        return i * 2
    ) ;; retorna 6
end
```

Expressões resultantes de avaliação de operadores

Indexação de ponteiros

A indexação de ponteiros devolve o valor de uma posição de memória indicada por um ponteiro. Consiste em uma expressão ponteiro seguida do índice entre parênteses rectos. O resultado de uma indexação de ponteiros é um *left-value*. Não é possível indexar ponteiros que designem funções.

Exemplo (acesso à posição 0 da zona de memória indicada por **p**): **p[0]**

Identidade e simétrico

Os operadores identidade (+) e simétrico (-) aplicam-se a inteiros e reais. Têm o mesmo significado que em C.

Reserva de memória

A expressão reserva de memória devolve o ponteiro que aponta para a zona de memória, na pilha da função actual, contendo espaço suficiente para o número de objectos indicados pelo seu argumento inteiro.

Exemplo (reserva vector com 5 reais, apontados por **p**): **[double] p = [5]**

Expressão de indicação de posição

O operador sufixo ? aplica-se a *left-values*, retornando o endereço (com o tipo ponteiro) correspondente.

Exemplo (indica o endereço de **a**): **a?**

Expressão de dimensão

O operador **sizeof** tem um único argumento e aplica-se a expressões, retornando a dimensão correspondente em bytes.

Exemplo: **sizeof(a)** (dimensão de **a**).

Exemplos e Testes

Os exemplos abaixo não são exaustivos e não ilustram todos os aspectos da linguagem.

Estão ainda disponíveis outros pacotes de testes.

O seguinte exemplo ilustra um programa com dois módulos: um que define a função **factorial** e outro que define a função principal.

Definição da função **factorial** no ficheiro **factorial.l22**:

```
public factorial = (int n) -> int:
  if (n > 1) then:
    return n * @(n - 1) ;; could also use 'factorial'
  else:
    return 1
```

Exemplo da utilização da função **factorial** no ficheiro **main.l22**:

```
;; external builtin functions
foreign int<> argc
foreign text<int> argv
foreign int<text> atoi

;; external user functions
use int<int> factorial

;; the main function
begin
  var value = 1
  write "Teste para a função factorial: "
  if (argc() == 2) then:
    text string = argv(1)
    value = atoi(string)
  write value, "! é ", factorial(value)
  return 0
end
```

Como compilar:

```
l22 --target asm factorial.l22
l22 --target asm main.l22
yasm -felf32 factorial.asm
yasm -felf32 main.asm
ld -melf_i386 -o main factorial.o main.o -lrts
```

Omissões e Erros

Casos omissos e erros serão corrigidos em futuras versões do manual de referência.

Categories: **Projecto de Compiladores** **Compiladores** **Ensino**